Problem 2: Transmission Lines (20 Points)

Consider the pair of coupled transmission lines shown below. The coupled section of each line has a near-end crosstalk coefficient k_{rx} of 0.1 and a far-end crosstalk coefficient, k_{fx} of 0. The aggressor line is driven directly by a 1V step source with a rise time of 100ps. The far end of the aggressor is terminated in a matched impedance. The victim line is shorted to the ground plane at the source end and terminated with a matched impedance at the far end. Note that the 'ground' symbol here denotes a local connection to the shared ground plane, not a single-point ground.



Using this information, sketch and dimension the voltage waveform at the far-end of the victim line (point H). You may ignore any effects that lead to waves with less than 10mV amplitude.

*At t=0ns (or 100ps, exactly), 0.5 V is launched on the aggressor line, and it induces cross talk at 1ns at B to F until it arrives at C, and then it is fully absorbed at the load at D with out any reflection. Thus the wave on the aggressor line induces cross talk to the victim line only between 1ns to 3ns (while it travels from B to C). On the victim line, the reverse coupling waves form a near end cross talk pulse of 2*2ns width on F, and it keeps traveling in reverse direction until it meets the shorted-end at E. This near-end cross talk wave is reflected at E, which will travel back up to H until it is absorbed at the matched termination. Make sure that we don't have forward traveling cross talk induced on the victim line because k_{fx} is 0. After all, what we have at point H is the reflected near end cross talk wave from point E, which has a magnitude of (-1)*0.1*0.5 = -0.05 V and a width of 4ns.*

Some students answered that there is no far-end cross talk at all at G and H (thus their potentials staying at 0 all the time). it is true that we have no far end cross talk because k_{fx} is 0, but we will get the reflected waveform of near end cross talk due to the shorted end at E.

And, some made a mistake that near end cross talk happening at F will propagate to G and H. This is not true at all because the near end cross talk 'always' travels in reverse direction and never in forward direction.

Also HSPICE simulation is performed for characterizing the cross talk waveforms, and the results and the source files are shown in the following pages. In this simulation, W-element is used for coupled transmission lines. If you want to have a look at the manual, refer to

http://www.ece.orst.edu/~moon/hspice98/ (chapter 21)

<Waveforms obtained by HSPICE>



<HSPICE files used in simulation>

1. ***SPICE input file***

```
*** Problem 2.
.param  z0 = 63.2456
Vin      in      0       pwl(0ns 0V 0.1ns 1v)
Rs       in      A       z0
Wa1     N=1  A       0       B       0
+         RLGCfile = simple_line.rlc
+         l=0.1581
```

```
Wv1    N=1     0       0       F       0
+          RLGCfile = simple_line.rlc
+          l=0.1581
Wc      N=2
+          B       F       0
+          C       G       0
+          RLGCfile = coupled_line.rlc
+          l=0.3162
 Wa2   N=1     C       0       D       0
+          RLGCfile = simple_line.rlc
+          l=0.1581
Wv2    N=1     G       0       H       0
+          RLGCfile = simple_line.rlc
+          l=0.1581
Rt      D       0       z0
Rtv    H       0       z0

.tran   10ps    11ns
.option post
.end
```

2. ***RLGC files***
1) ***coupled_line.rlc***

```
* This file defines the transmission line characteristics
* Note that it only concerns the transmission line part of the medium,
* and has nothing to do with parasitic inductors and capacitors shown
* on the schematic.

* The line properties are defined in terms of R, L, C, G terms.  The
* configuration is similar to Figure 6-10 of the textbook, except that
* it includes R and G factors.  Furthermore, skin effect and dielectric
* loss can be modeled through Rs and Gd terms.

* You don’t have to understand this file at all.  But basically, each
* of the parameters are specified by a lower triangular matrix.  you can
* think of diagonal terms as terms related to itself, and off-diagonal
* terms as inter-line characteristics.

* N = number of lines
*****************
2

* Lo = inductance matrix
********************
* L
* M     L
4e-7
8e-8    4e-7

* Co = Capacitance matrix
*********************
* Co+Cm
* -Cm         Co+Cm
10e-11
-2e-11   10e-11
```

* Ro = Resistance matrix
****
0
0 0

* Go = conductance matrix
****
0
0 0

* Rs = skin effect matrix
****
0
0 0

* Gd = dielectric loss matrix
****
0
0 0

2) ***simple_line.rlc***
* N = number of lines
1

* Lo = inductance matrix
4e-7

* Co = Capacitance matrix
10e-11

* Ro = Resistance matrix
0

* Go = conductance matrix
0

* Rs = skin effect matrix
0

* Gd = dielectric loss matrix
0